AutopilotMFD for Orbiter Flight / Space Simulation
Release Version 1.7



Description:

AutopilotMFD is a multipurpose MFD module for the Orbiter space simulation which implements a standard autopilot (AP) and flightplanning utility into every orbiter aircraft.

We are proud to be able to include the actual AIRAC database into the new version. Thanks to Richard Stefan from NAVData (www.navdata.at) for giving his permission. AIRAC contains all navigational aids available on earth as of September 2005 (VOR, NDB, ILS, Intersections(Fix)). AutopilotMFD 1.7 loads these data and displays them for navigation (see below). Furthermore, we have created an entry for every airport defined in the AIRAC database with correct location, runways, runways orientation, ILS Freq and runway width. Each airport has a provisional tower and a landing pad for refuel.

baed on AIRAC cycle	Airports	VOR	NDB	Intersections	ILSes
510	10003	~4000	~6000	~78000	~ 2000

All airports are defined in the new earth.cfg (CONFIG folder) and ICAO.cfg files are in the subdirectory /CONFIG/AIRAC. Expect longer orbiter loading times for earth scenarios. BACKUP the old earth.cfg file before download and installation !

The autopilot has the 1.51 autopilot panel (see above) which should be quite familiar to flightsimmers and easier to operate. In 1.7, we have additional reading fields on the AP panel:

Left entry (ICAO) is a target type selector field - click right / left portion of the field to toggle between the possible targets.
Once selected a target type, click the middle field (EDDM) to toggle through the (filtered) available targets of that type (ICAO) in range. Right field: Distance to selected target (DME).

When you open the AP panel, there will be a second panel available - the radar or EFIS panel.

This displays all navaids relative to your vessels position. On the Fig above, you are DG GL-01S (red triangle on bottom of EFIS). The white dotted line is your heading. You are heading for Munich airport (EDDM), Germany. Airports are displayed in PINK with ICAO name and Fullname, VOR's in GREEN with a small rectangle, NDB's in blue with a small circle. Fixes are WHITE with a triangle. Selected targets have a YELLOW circle.

Notice the range button right upper corner of the EFIS. Click right / left of the button to toggle through the available ranges (5 km to 160 km). The Fig. above shows an overview of the munich area, the DG heading southwest into Salzburg VOR. Selecting targets from the EFIS is very simple: Click onto the target (f.e. SBG VOR), press NAV button on the AP panel, and the vessel will turn into the correct heading.

Overview of EDDM with ILS navaids on the EFIS.

On higher EFIS ranges, the display can become quite crowded.
Press CTRL key on your keyboard and click anywhere on the EFIS screen to enable / disable the display of FIXES:

Press the SHIFT key on your keyboard and click anywhere on the EFIS panel to enable / disable the display of airports:



Fig : Final approach on RW 08l EDDM

1.7 **supports VOR, NDB, ILS** and **GPS navigation**. Define your own GPS fixes in /flightplans/gpsdef_earth.cfg.

AP targets are now **range-sensitive** - VOR's and NDB's have a 300 km range, GPS fixes generally have 100 km, ICAO airports have 500 km, orbiter bases have 3000 km.

1.7 allows for precision approaches (simulated ILS) and, in flightplans, **autolanding** on runways. **Ground taxi** (not modelled in orbiter) was implemented by an AddForce hack found in the forum. On moon, 1.7 supports 'thruster only' vessels like **landers** (TTM24H PTM + lander or rover, LLRV-V2). For hovering on low gravity bodies like moon 1.6 now offers an enhanced vector mode for easy vessel navigation.

Test flight I, Earth

I suggest to get used to the AP with a standard Deltaglider on earth (KSC). Enable the AutopilotMFD in the module section of Orbiter's launchpad. Call up the **standard orbiter situation file** (Deltaglider -> Ready for takeoff KSC). Switch to the main panel by pressing F1 and F8, F8. Call up the AutopilotMFD (Fig. see below) and activate the new autopilot panel (Fig. see above). This panel will be available to you throughout your cruise and also in external views. PLEASE remember to set 'focus' to Orbiter's main window (click on main window first) when you want to make keyboard or joystick input to your plane. Alternatively, you may 'minimize' the AP panel before grabbing the joystick.

Assuming earth and on runway KSC, push the AP main button. Button should now be lit red. Set desired IAS by clicking into the left/right half of the IAS/M reading. By clicking CTRL or SHIFT and left mouse button you can modify the rate of increase / decrease in your reading windows. Set desired altitude (click right/left next to the altitude def button). Set desired climb rate (click on the upper / lower part of the VS slider button). Press the SPD button - aircraft is now rolling.
At V2 speed, press the ALT button and the LVL button - AC is climbing.
Define an initial heading by pressing left/right next to the DEF heading button (see above) and press HDG. AC is turning into the heading.
Press the MACH button. IAS reading changes to MACH. Define the cruise speed by right/left clicking into the IAS/M reading.
Remember - all this only works of the Autopilot main button is **activated** !
You may, however, preselect the AP switches, altitude etc when AP main switch is off.

Look back to the Autopilot's MDF panel.



Press Ob+ or Ob- to toggle through all targets available. Or use the EFIS panel (see above) for navigating. Select a target and press 'NAV' on the AP main panel to turn the AC into the target heading. By pressing PD+ and PD- on the MFD you can toggle through available pads at the base. If you select another vessel as a target, you can toggle the docking ports. Docking itself is NOT supported yet.

Set descend rate and altitude to initiate descend and press the ALT and the LVL button again. Manually land the AC.

For **ILS landing**, proceed as follows: Navigate to a point from where you can intercept the ILS easily. This is usually at about 1500m altitude heading approx. 30 degrees crossing the ILS localizer course (see below).



On the AP panel, select ICAO on the target type selector. On the target selector, select the correct airport (assuming Munich, we would choose EDDM). Then select the correct target type (ILS) on the target type selector. The ILSes are now filtered for your airport. Choose the right ILS. For Munich (EDDM) arriving from the west this would be either IMNE (runway 08L) or IMSE (runway 08R). Press NAV on the AP panel. The vessel will turn into the localizer shortly before crossing the Loc-course and follow the glideslope. This is a MARK I implementation - expect more precise implementation with full autolanding in one of the next versions.

This figure shows the selected ILS on the AP panel (IMSE). On the MFD, you have the correct ILS frequency, the standard localizer course (82.0 deg) and the usual glideslope for this ILS.

You can tune this frequency into the NAV frequency of orbiter by hand - but pls do not expect that orbiters ILS indication and the AP match - it is impossible to link to the NAV MFD due to API limitations.

On final, the AP will flare the vessel and land. Wheelbrakes are operated until the vessel has a groundspeed of 15 m/s.

Test flight II, Moon

To test the hover functions, please call up the standard deltaglider scenario Brighton Beach (Folder Deltaglider, -> Brighton beach). Jump into the cockpit by pressing F1 and F8, F8. In the Deltaglider main panel, OPEN THE RETRO THRUSTER COVERS (a switch somewhere on the right side of the panel) to enable the retro engines working !

Call up the Autopilot MFD. Call up the autopilot main panel as above.
Set desired hover altitude and press T/O-button or set climb rate and press ALT und LVL-buttons. AC should now be climbing with hover engines and level out at target altitude. When the AC is stable at desired altitude, press LND button. The AC will descend with a altitude-adjusted sink rate and finally land on ground. Press T/O button again after setting the target altitude. Set a heading and press HDG button or select a target (look at the MDF and press Ob+ or Ob- buttons to select the target) and press the NAV-button on the AP main panel. The vessel will turn into the target. Press VCT button after HDG or NAV to **turn the speed vector** into the heading or target. When in VCT mode, the vessel will **accelerate/ decelerate to the speed defined as set in IAS field**. AC should be turning. Or simply engage the main thrusters to accelerate.

In case you need a full hover 'stop' of the AC over ground, press the stop button (arrow down on the AP main panel) and wait until AC is in a complete stop.
500 km inbound the target base decrease speed to 500 m/s. Select a landing pad by pressing the Pa+ or Pa- buttons on the MFD or press PaF to select a free pad. Navigate directly to the pad and land manually or press **LND button (first) and then VCT button (second)**. AC will turn onto the correct heading, control approach speed and autoland.

Tips:

- Ground taxi on earth: Call up the AP panel, press AP on, select IAS 15 m/s, press SPD button (vessel accelerates), select a taxi heading in the heading reading field, press HDG button.
- Moon: Press the ALT and the HLD button anywhere to maintain correct altitude.
- Moon: In case you hover somewhere and want to land, simply press the LND button to call up the landing program. This will control sink rate automatically and you can approach the pad easier.
- Altitude and VS: You can switch between **meter** and **feet** units display in AP's main panel by CTRL+leftclick on the ALT button. This will also change the VS indication to ft/minute
- Speed: You can switch between **meter/ sec** and **knt** units display in AP's main panel by CTRL+leftclick on the SPD button.
- Clicking the thruster/airfoil control (see Fig 1) switch will toggle the mode of operation of the autopilot. If "air" (default on earth) aerodynamic rudder control is enabled (no hover engines and thrusters). If "thr", thruster control, hover engine and no rudders are used. Earth: If you climb above 12 km into very thin air switch to "thr" ...
- elevator trim can be directly modified from the AP panel.
- Increment and decrement orbiters time-acceleration in single steps by pressing the T+ and T- buttons on the MFD

Further remarks:

- Windows 98 is fully supported (you will need a patch - please visit the Orbiterforum/Addons) and search for Windows 98 and 1.7
- if one of the AutopilotMFD scenarios (orbiter launchpad, AutopilotMFD folder) is started, the AC are usually in ***flightplan mode***. If in FP mode, the indication 'FP'

on the AP main panel (top, left) is lit red. In this mode, clicks to the AP panel (not the MFD) are blocked for one exception: AP main button. Click this once to finish flightplan mode (FP indication will be off) and resume normal operation with a responsive AP panel.

- Some of the supplied scenarios need third party addons. If these are not installed in your system, the scenario will fail. The addon modules required (all excellent, find them at www.orbithangar.com): New Eagle (heavy freighter), Moonbase Alpha (moon addon scenery), Shado base (another small moon base). These addons are **not** required if you use AutopilotMFD in normal orbiter operation.

Each spacecraft in the simulation initiates an independent instance of autopilot which monitors and controls **all** spacecrafts in the simulation. This is a complex task, and involving many aircrafts may consume hardware power. AutopilotMFD has been primarily tested with the standard Orbiter aircraft's.

Installation:

- BACKUP your earth.cfg file (/CONFIG folder)
- Extract the ZIP file into the \Orbiter home directory
- Enable the MFD under 'Modules' in the Orbiter Launchpad

Limitations: AC should have movable rudders to be controlled by AP on earth. If they have not, try to switch to "thr" mode (see above) before takeoff.

Credits: Many thanks to Richard Stefan from NAVData for providing his perfect selection of NAVAIDS and the permission to publish those to the orbiter community for free. AP panel similar to the panel of the freeware FS2004 Gulfstream IV modelled by Roger Mole (flightsim@ukgo.com).

Have fun and await further enhancements.

The AutopilotMFD team

Developed with Visual C++ 6.0 and the OrbiterSDK. Requires Orbiter 2005. May be freely copied for private use. Commercial use is prohibited. Contacts or comments welcome to info@scientific-networks.de or post at orbiterforum, addons

Abbreviations:

AC	aircraft
AIRAC	Aeronautical Information Regulation and Control
ALT	altitude autopilot main button
AP	autopilot
DAFIF	Digital Aeronautical Flight Information File
EFIS	Electronic Flight Instrument System
FIX	navigational waypoint
FL	flight level (cruise altitude)
FP	flight plan
ft	feet (altitude unit)
HDG	heading
HLD	hold altitude button (main panel)
IAS	indicated airspeed
ICAO	International Civil Aviation Organisation; here: standard abbr. for airports
INTERSECTION	navigational waypoint usually defined by VOR radials
ILS	Instrument landing system
knt	knots (speed unit)
LVL	level altitude button (main panel)
m	meters
m/s	meter per second
MFD	multipurpose function display
NAV	navigation button on AP panel
NAVDATA	Navigation update service for flightsimulation c:(www.navdata.at)
NDB	Non-Directional Beacon
QNH	Barometric pressure
VOR	VHF Omni-Directional Range
V2	IAS at which the AC can take off
VS or V/S	vertical speed

MFD Buttons:

Function	Key shortcut Right or left <shft>	Description
<Pg>	<shft> p	Toggle MFD pages. Page three will show your actual flightplan. The current command of your flightplan is highlighted.
<Cmd>	<shft> c	Open the command window from where you can enter the commands (listed in second table below) or load and start a flightplan
<Obj+>	<shft> o	Toggle forward through all available target objects in the simulation. Spaceports will come first, then all spacecraft and stations.
<Obj->	<shft> i	Toggle backword through all available target objects in the simulation. Spaceports will come first, then all spacecraft and stations.
<Pan>	<shft> a	Open Autopilot main panel
<Pa+>	<shft> r	select next landing pad of selected base
<Pa->	<shft> t	select prev. landing pad of selected base
<PaF>	<shft> l	select a free landing pad
<T+>	<shift> h	Increment orbiters timeacceleration single step
<T->	<shift> g	Decrement orbiters timeacceleration single step
<Stp>	<shft> s	Stop all autopilot functions by the click of a button

Manual command line input / flightplan commands

[ ] abbrevation
<num> is number (float)
APA: Autopilot for spacecraft orientation.
APH: Horizontal autopilot
APV: Vertical autopilot
FP: Additional flightplan / command line arguments

Group	Arguments	Type	Description
aileron [ail]	left <num>	**APA**	set aileron rudder left <num> %
aileron [ail]	neutral	APA	explains itself
aileron [ail]	right <num>	APA	set aileron rudder right <num> %
bank	- or + <num>	APA	bank vessel to <num> positive or negative degrees . Once started, this function will bank until the stop apa command is used.
burn	retro <num>	APA	burn retro engines at level <num> % ; use wait function and then 'stop engines' to finish the burn. You can wait for a time (s), an altitude (m), a climb or a sink rate (m/s).
elevator[elv]	down <num>	APA	set elevator rudder down <num> %
elevator[elv]	neutral	APA	explains itself
elevator[elv]	up <num>	APA	set elevator rudder up <num> %
flaps [flp]	down <num>	APA	set flaps down (in) <num> %
flaps [flp]	up <num>	APA	set flaps up (out) <num> %
flaps [rud]	neutral	APA	explains itself
head	<num>	APA	fix a permanent vessel heading in a flightplan (use stop apa) to discontinue. This is different from 'turn heading', because 'turn heading' stops when heading is achieved. Under atmospheric conditions, rudder and airleon are both operated for the turn.
headtarget		APA	fix a permanent target heading. Use stop apa to discontinue ...
pitch	- or + <num>	APA	pitch up and down <num> degrees
rudder [rud]	left <num>	APA	set rudder left <num> %
rudder [rud]	neutral	APA	explains itself
rudder [rud]	right <num>	APA	set rudder right <num> %
stop	apa	APA	stop vessel angle (pitch / bank / head) autopilot. The only way to stop the autopilot's bank, head or pitch function. This is handy because you can simultaneously run other tasks: Pitch up to 40 degrees, then burn main engines and wait for events to occur, with stable 40 degrees pitch ....
run	aph	**APH**	run the horizontal autopilot.
run	hold	APH	bring the vessel to a near complete stop over ground. Ends in anterograde speed of 0.2 m/s always. If initial speed > 200 m/s, vessel will turn retrograde and brake with main thrusters.
run	land	APH	automatically hover-land the spacecraft. This refers to vertical control only. Horizontal control must be either programmed in fliphtplan (see examples) or manual.
run	speed <num>	APH	Accelerate /decelerate vessel to <num>

			m/s. Before speeding up, you must turn (see below) into the appropriate direction or towards a target. Use this also for braking if <num> is less than your current speed. If speed-diff is > 200 m/s vessel will be turned retrograde and main thrusters will be activated. Otherwise retro thrusters are used. In vessels without RETRO thrusters ship will always turn retrograde if speeddiff is > 5 m/s
run	target [tgt] vector [vct]	APH	this autopilot mode will correct the current speed vector in a way that it will exactly hit the target. This implies that the speed vector is approximately correct before starting this mode. If the speed vector differs considerably, the ship is turned +- 90 degress and main engines are fired. If you are close to a target (say 3000 m) pls rather exactly speed up to the target (f.e. pad) otherwise the vessel will not hit it.
run	target [tgt] vectorland	APH	autopilot will compute the descend rate to a given target (select before, f.e. pad on spaceport) and bend the velocity vector and the sink rate to exactly hit the pad.
run	target [tgt] vectorspeed [vctspd]	APH	runs target vector program and, in addition, will brake in between so that the vessel will come to a near complete stop just close to the target.
run	target [tgt] vectorspeed [vctspd] <num>	APH	see above, but will stop in <num> m (f.e. 1000 m) hold distance inbound the target. From there, you can start the final approach. If you properly selected a pad (see above) as a target you can stop at f.e. 30 m and enter "run land" afterwards. This will bring you, out of 100 m altitude (adjust before), almost certainly onto the pad. You may want to run and have a look at the flightplan examples.
stop	aph	APH	stop the horizontal autopilot
turn	anterograde	APH	turn vessel horizontally levelled into anterograde direction (into the speed vector). This is different to orbiter's PROGRADE, because the vessel will only change the heading, not the bank angle.
turn	current <curr> object <obj>	APH	turn the vessel horizontally levelled exactly onto the selected target
turn	heading <num>	APH	horizontally turn the vessel to <num> degrees. Ship is always h-levelled against the planetary surface.
run	apv	**APV**	start the vertical autopilot. Set the target altitude and the climb/descend rate before.
run	takeoff	APV	takeoff and climb to 30 m over ground; hold there
run	takeoff <num>	APV	takeoff and climb to <num> meters
set	altitude [alt] <num>	APV	preselect target altitude (m)
set	climb [desc] <num>	APV	preselect climb / sink rate (m/s)

set	climb [desc] autoclimb [ac]	APV	preselect autoclimb; set climb rates automatically
set	climb [desc] autodescend [ad]	APV	preselect autodescend; set descent rates automatically
set	climb wp	APV 1.6	set the climb rate to a value that the predefined altitude is hit at the next waypoint
stop	apv	APV	stop the vertical autopilot
'		**FP**	This starts a comment line in your flightplan
//		FP	This starts a comment line in your flightplan
block	timeacceleration [timeacc]	FP	with this command you can block other spacecraft in the sim to accelerate time. Use deblock to reverse.
burn	main <num>	FP	burn main engines at level <num> % ; use wait function and then 'stop engines' to finish the burn. You can wait for a time (s), an altitude (m), a climb or a sink rate (m/s).
burn	retro <num>	FP	burn retro engines at level <num> %
burn	hover	FP	burn hover engines at level <num>%
deblock	timeacceleration [timeacc]	FP	
end		FP	always the last line of flightplan
load	flightplan <fp> <name>	FP	load flightplan <name>. A file <name>.txt (your flightplan) must exist in the subdirectory "flightplans" in the orbiter main directory.
loop		FP	restart the flightplan at item 1
loop	<num>	FP	restart the flightplan at item <num>
loop	<num1> <num2>	FP	restart the flightplan at item <num1> for <num2> times
maintain	speed	1.6	useful on earth. Will control speed as defined by a 'set speed' command throughout the flightplan.
REM		FP	This starts a comment line in your flightplan
resume	flightplan	FP	restart the flightplan at the current item position
run	antinormal	FP	turn vessel orbit ANTINORMAL (orbiter inbuild)
run	flightplan <fp>	FP	run the current flightplan from item 1. ONLY to be used from autopilot's command line, not within a flightplan.
run	hlevel	FP	run orbiters HLEVEL function, this way it can be used in a flightplan
run	holdalt	FP	run orbiters HOLDALT function
run	killrot	FP	run orbiters KILLROT function, so it can be used in your flightplan
run	normal	FP	turn vessel orbit NORMAL (orbiter include)
run	oretrograde	FP	turn vessel orbit RETROGRADE (orbiter inbuild)
run	prograde	FP	turn vessel orbit PROGRADE (orbiter inbuild)
select	<name>	FP	select target object (vessel or spaceport)
select	freepad	FP	select next free pad in spaceport. Pls select a valid spaceport before that. Pad is free when no spacecraft has landed on it or when no other vessel has reserved the pad (called freepad before)
select	pad <num>	FP	select pad number <numb> of current

			spaceport regardless of free or not
set	attmode <rotation> <linear> <off>	FP	set attitude mode in current vessel
set	afcontrol <on><off>	FP	atmospheric rudder control by user on or off. Set off if your flightplan wants to take the stick.
set	apmode thruster	FP	set autopilotcontrol to thruster
set	apmode airfoil	FP	set autopilot control to rudder
set	timeacceleration [timeacc] <num>	FP	speed up or reduce time acceleration within a flightplan
stop		FP	stop all autopilot functions
stop	all	FP	stop all autopilot functions and the flightplan
stop	antinormal	FP	orbiters ANTINORMAL off
stop	flightplan	FP	stop the current flightplan
stop	hlevel	FP	turn orbiters HLEVEL off.
stop	holdalt	FP	orbiters HOLDALT off
stop	killrot	FP	orbiters KILLROT off
stop	normal	FP	orbiters NORMAL button off
stop	oretrograde	FP	orbiters RETROGRADE off
stop	prograde	FP	orbiters PROGRADE off
turn	retrograde	FP	turn vessel into RETROGRADE direction
vectorheading		1.6	bends the speedvector onto the defined heading
vectortarget		1.6	bends the speed vector into the selected target
wait	<num>	FP	wait for <num> *seconds*, then resume flightplan
wait	abs <num>	FP	wait until *second* <num> after start of orbiter. After that, resume flightplan. F.e. takeoff with other spacecrafts, and "wait abs 120" -> 2 minutes for until all vessels are in the air. When all have the same flightplan (see below) and at target altitude, you can organize a nice fly-off from a spaceport.
wait	altclimb <num>	FP	wait until vessel is climbing above <num> meters
wait	altdesc <num>	FP	wait until vessel is descending to less than <num> meters
wait	anterograde	FP	wait in flightplan until vessel is anterograde
wait	climbrate <num>	FP	wait in flightplan until vessels climbrate is <num> m/s or more. This is mainly to be used with 'burn main' function (see above).
wait	distin <num>	FP	wait in flightplan until the vessel is less than <num> meters inbound the target
wait	distout <num>	FP	wait (stop flightplan processing) until the vessel is distant <num> m outbound the current target
wait	freepad	FP	wait until we have a free pad in the current spaceport. :-) Send f.e. ten eagles to a spaceport and let them land one after the other. See and run the demo Moonbase Alpha flightplan for an examples.
wait	retrograde	FP	wait in flightplan until vessel is retrograde
wait	sinkrate <num>	FP	wait in flightplan until sinkrate is <num> m/s or less. Example: use this in a ballistic flightplan. Burn main engines for a specified

			time, then turn engines off, and wait until gravity of the planet / moon brings the vessels sink rate to less than -1.0 m/s. From there you can continue your plans (f.e.stabilize your orbit, perform a retrograde burn, etc).
wait	speedless <num>	FP	wait in flightplan until speed is less than <num> m/s
wait	speedmore <num>	FP	wait in flightplan until speed is more than <num> m/s. Use this in combination with burn main command
wait	inrange <name>	new 1.6	waits until <name> target (VOR, GPS, Airport) is in range
wait	intersection <num>	new 1.6	wait until we have an intersection <num> degrees with the currently defined target
waypoint	<alt> <spd> <name>	1.6	simply define VOR, GPS, airport waypoints. 'waypoint 1500. 220. DME001' will make the vessel: (1) adjust climbrate to hit 1500 m altitude at 'DME001' (a GPS fix). will set speed to 220. Will wait in flightplan until: (air) 1000 m inbound the target or (ground) 30 m inbound the target. This is all in one command.
flaps	up <num>	FP	set flaps up by <num> %
flaps	down <num>	FP	set flaps down by <num> %
flaps	neutral	FP	
operate	gear	FP	gear up or down
operate	dockingport	FP	docking / undocking
operate	rcoveropen	FP	open the retro covers if any
operate	rcloverclose	FP	close them
operate	wheelbrake	1.6	....
release	wheelbrake	1.6	....

Digital autopilots: What is the theory behind this ?

The actual version involves a new AP engine which controls attitude, retro, main and hover thrusters as well as rudders by a PID algorithm. PID stands for proportional, integral, and derivative. Typically a PID controller manipulates one control output to force a process value towards a reference point. We have a good basic summary of the technique (flightgear) here:

http://www.flightgear.org/Docs/XMLAutopilot/node2.html

Programming this type of control is a complex task and, if many vessels involved, can be hardware-intense. In real, we have autopilots usually been developed for one vessel type only. Please remember that if the current version fails ....

Last but not least, Orbiter has a very good API available, which makes things often easier as expected. .... Tnx, Martin !

Automatically start a flightplan in an orbiter scenario

The AP enables you to load and run this flightplan to one or many spacecrafts at a given spaceport. Please open the scenario file in a text editor (in orbiters subdirectory "scenarios") and enter the following lines at

```
BEGIN_MFD Right
  TYPE User
  MODE AutopilotMFD
  PAGE 0
  FLIGHTPLAN PB-01 fp_brighton_pb 2 0
END_MFD
```

This loads the Autopilot module into the right MFD, and in this example, the vessel "PB-01" will load the "flightplan fp_brighton_pb" after 2 seconds and start at item 1 (position 0) of the flightplan. *Do not use spaces in the flightplan vessel names*. Define your vesselnames without spaces.

Orbiter's time acceleration

In this version of the software it is possible to control Orbiter's time acceleration via a flightplan command. In waiting periods of your flightplan, use 'set timeacc <num>' to accelerate the simulation, and before a timecritical period in your flightplan use 'set timeacc 1' to set timeacceleration back to normal. You can always override the acceleration setting inflight by user input. For debugging your flightplan it can be useful to enter a 'set timeacc 0.1' command at the 'buggy' flightplan position of your choice. This avoids having to watch the entire flight ...

Save current scenario

When you exit a scenario in which AutopilotMFD has been used, the MFD will save basic MFD parameters (which autopilot is running, etc.) to disk. In particular, the active flightplan and the flightplan position will be saved.

Flightplan monitoring

The actual flightplan as well as the currently active step in flightplan can be checked inflight on page 3 of the MFD.

1.6: New flightplan logic

Please note: A flightplan can have up to 1024 items (excl. comments). On earth (atmosphere), the flightplan demon will continue in the current flightplan until a valid wait command is reached. Will then continue as soon as the wait condition is fulfilled. On moon, the flightplan demon will switch to the next item of the flightplan one by one. as soon horizontal (aph) AND vertical (apv) autopilot are OFF (the previous command has been done) and no valid wait command is in use. The vessel axis autopilot (apa) which can be used for individual vessel orientation, will keep on working throughout the flightplan unless it is explicitely stopped by 'sop apa' command. Please see the more sophisticated flightplans in the flightplan folder.
Also note: Define your wait's in a way they can be achieved. Heavy vessels like the DGEX may not turn into a target (GPS FIX) in time.

GPS definitions

Find some GPS definitions in the file **gpsdef_earth.cfg**. This is located in the /flightplans/ subdirectory. Once defined here, the definitions will be loaded upon each simulation start and in all eatrh flightplans. The format is:

[Name] [LON] [LAT] [RAD] ; [Longname optional]

An example for KSC (inbuild 1.6) here:

```
BEGIN_GPSFIX
RW33START -80.6859 28.5907 0. ;  FIXES Cape Canaveral Earth
RW33END -80.7087 28.6253 0.
TAXI01 -80.6982 28.6277 0.
TAXI02 -80.6953 28.6293 0.
TAXI03 -80.6813 28.6077 0.
TAXI04 -80.6758 28.5989 0.
TAXI05 -80.6654 28.5866 0.
TAXI06 -80.6589 28.5804 0.
TAXI07 -80.6587 28.5555 0.
TAXI08 -80.6669 28.5212 0.
TAXI09 -80.6724 28.5203 0.
TAXI10 -80.6750 28.5203 0.
RW33HOLD -80.6842 28.5909 0.
DM001 -80.7904 28.7531 0.
DM002 -80.6670 28.4536 0.
DM003 -80.5799 28.3042 0.
DM004 -80.5294 28.3382 0.
DM005 -80.5735 28.4194 0.
DM006 -80.6304 28.5060 0.
END_GPSFIX
```

You can add own GPS headings here and/or post them or email them (TAXI pls, there should be enough airborne fixes available) to be included in one of the next AutopilotMFD versions.

Flightplan Waypoints

It has never been easier to define an atmospheric flight in Orbiter. Usually by hand, one would select a target, select an altitude, select a speed, and select a waiting condition when to switch to the next waypoint. In the current version of the AP, we have a command which does all this in one line: waypoint.

waypoint [ALT] [IAS] [NAME]

Some KSC VOR's here:

waypoint 5000. 250. OMN
waypoint 8000. 250. CRG
waypoint 8000. 250. VQQ
waypoint 8000. 250. GNV
waypoint 5000. 250. ORL

and some Taxi instructions here:

waypoint 0. 3. TAXI10
waypoint 0. 5. TAXI09
waypoint 0. 15. TAXI08
waypoint 0. 20. TAXI07
waypoint 0. 15. TAXI06
waypoint 0. 15. TAXI05

Are there VOR's on the moon ;-) ?
Not yet. Maybe Martin defines some in moon.cfg ....

Sample flightplans

You find several flightplans for the AutopilotMFD in the subdirectory 'flightplans' of the orbiter home directory. Edit them with a conventional text editor.

EARTH demo flightplan for a DG from Munich to Frankfurt, Germany (EDDM to EDDF), see sample scenario

```
// ------------------------------------------------------
// Flightplan for the orbiter space simulation
// for use with the Autopilot MFD
//
// AutopilotMFD (c) 2005 by R. Bumm, Munich
// ------------------------------------------------------
// small flightplan routine (Deltaglider) for takeoff,
// VOR and GPS navigated flight from Munich RW 26R to Frankfurt RW 25R
// using the GIVMI SID and a PSA arrival
//
set apcontrol airfoil
set alt 500
set climb 5.0
run speed 220.
wait speedmore 170.
run apv
//                  continous speed control througout this flight
maintain speed
wait altclimb 50.
//                  gear up
operate gear
waypoint 1000. 220. DM060
waypoint 5000. 250. DM063
waypoint 8000. 250. GIVMI
waypoint 8000. 250. TALAL
waypoint 10000. 250. ALB
waypoint 10000. 250. NORAS
waypoint 8000. 220. WUR
waypoint 3000. 220. PSA
waypoint 1500. 200. CHA
waypoint 1500. 200. HNU
select REDGO
wait distin 5000
operate gear
select IFNW
end
```

EARTH demo flightplan incl. VOR navigation, autolanding, ground taxi, and refuel (run the VOR demo scenario for two DG's with this flightplan):

```
// ----------------------------------------------------------
// Flightplan for the orbiter space simulation
// for use with the Autopilot MFD
//
// AutopilotMFD (c) 2005 by R. Bumm, Munich
// ----------------------------------------------------------
// small flightplan routine (Deltaglider) for takeoff, VOR and GPS navigated flight, and
landing at KSC RW33
//
set apcontrol airfoil
set alt 500
set climb 5.0
run speed 220.
wait speedmore 170.
run apv
//                  continous speed control througout this flight
maintain speed
wait altclimb 50.
//                  gear up
operate gear
//                waypoint: target altitude at waypoint, IAS, and waypoint name
//                waits until 1000 m inbound the target
//                DM001 is a fix north of the RW33, reach it at 1000, 220 m/s
waypoint 1000. 220. DM001
//                along the coast , Ormond Beach (OMN)
//                further north, Craig (Jacksonville) VOR (CRG)
//                now west, Cecil VOR (VQQ)
//                now south, Gators VOR (GNV)
//                and to: Orlando (ORL), and begin descend
waypoint 5000. 250. OMN
waypoint 8000. 250. CRG
waypoint 8000. 250. VQQ
waypoint 8000. 250. GNV
waypoint 5000. 250. ORL
//                further descend back home
waypoint 1500. 220. Cape Canaveral
//                into left circuit RW 33
waypoint 1500. 220. DM002
waypoint 1500. 200. DM003
select DM004
wait distin 3000.
set alt 0.
set climb -10.
set speed 220.
select DM005
wait distin 1000.
//                  gear down
operate gear
//                  simulated 'ILS' landing
select RW33START
set alt 10.
set desc wp
wait distin 3000.
set speed 190.
```

```
head 330.
wait altdesc 20.
//                  flare ...
set alt 0.
set climb -2.0
set speed 150.
wait groundcontact
//                  avoid 'jumps' post landing
elv down 20.
set speed 15.
operate wheelbrake
wait speedless 20.
elv neutral
release wheelbrake
set speed 15.
//                  taxi circuit
waypoint 0. 15. RW33END
waypoint 0. 15. TAXI01
waypoint 0. 15. TAXI02
waypoint 0. 15. TAXI03
waypoint 0. 15. TAXI04
waypoint 0. 15. TAXI05
waypoint 0. 15. TAXI06
waypoint 0. 15. TAXI07
waypoint 0. 15. TAXI08
waypoint 0. 15. TAXI09
select TAXI10
wait distin 80.
set speed 0.
operate brakes
wait 10
select Cape Canaveral
//                  wait is important to receive a free pad
wait 4
select freepad
release brakes
set speed 5.
wait distin 10.
set speed 0.
operate brakes
//                  you must stop the ap speed engine to enable
//                  to shutdoen all the engines and enable refuel
stop aps
burn main 0.
wait 3
stop engines
wait 10
run speed 2
maintain speed
waypoint 0. 3. TAXI10
waypoint 0. 5. TAXI09
waypoint 0. 15. TAXI08
waypoint 0. 20. TAXI07
waypoint 0. 15. TAXI06
waypoint 0. 15. TAXI05
select RW33HOLD
wait distin 80.
set speed 0.
```

```
operate brakes
wait 30.
release brakes
set speed 5.
select RW33START
wait distin 10.
head 330.
wait yawmore 329.
set speed 0.
operate brakes
wait 10.
release brakes
loop
end
```

MOON demo flightplan (run the DG for Moon orbit scenario):

```
// ------------------------------------------------------
// Flightplan for the orbiter space simulation
// for use with the Autopilot MFD
//
// AutopilotMFD  (c) 2005 by R. Bumm, Munich
// ------------------------------------------------------
// flightplan routine for reaching a stable moon orbit
// and deorbit to Brighton Beach.
// Version for Deltaglider III
// select home base
select Brighton Beach
//  take off to 50 m
// notity use of the 'and' in the line below; will run takeoff and turn simultanously
// works only if you 'and' vertical and horizontal autopilot functions
run takeoff 50 and
turn heading 270
//              lets go west ....
run killrot
wait 3
stop killrot
operate gear
burn main 10.
//              pitch, bank and head commands will work simultaneously
//              in flightplan
//              until we stop apa (see below)
pitch 40.
wait 10
burn main 100.
wait altclimb 5000.
pitch 20.
wait altclimb 10000.
pitch 10.
wait altclimb 20000.
pitch 0.
wait altclimb 33000.
stop apa
stop engines
run prograde
wait anterograde
set timeacc 100.
wait sinkrate -0.2
//              timecritical so:
set timeacc 1
burn main 100
wait speedmore 1560
//              lean back, now in stable orbit .....
stop engines
// in orbit
set timeacc 100
wait 3000.
wait inrange Brighton Beach
select Brighton Beach
wait distin 2500000.
set timeacc 1
run hlevel
turn retrograde
```

```
burn main 100.
wait speedless 1400.
stop engines
run hlevel
turn anterograde
run hlevel
wait distin 450000.
set timeacc 1
turn retrograde
run hlevel
set timeacc 1
burn main 100.
wait speedless 800.
stop engines
set alt 3000.
set climb ad
run apv and
run tgt vector and
wait distin 48000.
//          IMPORTANT to stop the horizontal autopilot (aph) here
//          otherwise flightplan will hang. Will not read next step
//          until aph and apv are done !
stop aph
set timeacc 1
run hlevel
run hold
select Brighton Beach
//          wait until we have a free landing pad
wait freepad
turn current obj
run killrot
wait 5
//          approach pad
run speed 50.
operate gear
run target vectorland
//          and wait for refueling ... :-)
wait 20.
loop
end
```